Solving Recurrence for ODE of a circle

Explicit Euler

```
> restart:
```

Approximate x'=-y, y'=x with Explicit Euler

```
> REC1 := x(k+1)=x(k)-h*y(k) ;
  REC2 := y(k+1)=y(k)+h*x(k) ;
  INI  := x(0)=1, y(0)=0 ;
```

$$REC1 := x(k+1) = x(k) - h\, y(k)$$

$$REC2 := y(k+1) = y(k) + h\, x(k)$$

$$INI := x(0) = 1, y(0) = 0 \quad \textbf{(1.1)}$$

```
> SOL := simplify(rsolve( {REC1,REC2,INI},{x,y}));
```

$$SOL := \left\{x(k) = \frac{1}{2}\left(\frac{h^2+1}{1+\mathrm{I}\,h}\right)^k + \frac{1}{2}\left(-\frac{h^2+1}{-1+\mathrm{I}\,h}\right)^k, y(k) = \frac{1}{2}\,\mathrm{I}\left(\left(\frac{h^2+1}{1+\mathrm{I}\,h}\right)^k - \left(-\frac{h^2+1}{-1+\mathrm{I}\,h}\right)^k\right)\right\} \quad \textbf{(1.2)}$$

Check the discrete "prime integral" x^2+y^2=C

```
> SOLINT   := simplify(subs(SOL,x(k)^2+y(k)^2)) ;
  L        := op(1,op(1,SOLINT));
  R        := op(1,op(2,SOLINT));
  SOLINT2  := simplify(L*R);
  SOLINT3  := (1+h^2)^k ;
```

$$SOLINT := \left(\frac{h^2+1}{1+\mathrm{I}\,h}\right)^k \left(-\frac{h^2+1}{-1+\mathrm{I}\,h}\right)^k$$

$$L := \frac{h^2+1}{1+\mathrm{I}\,h}$$

$$R := -\frac{h^2+1}{-1+\mathrm{I}\,h}$$

$$SOLINT2 := -\frac{\left(h^2+1\right)^2}{-h^2-1}$$

$$SOLINT3 := \left(h^2+1\right)^k \quad \textbf{(1.3)}$$

```
> X := seq( Re(subs(SOL,h=Pi/40,x(k))),k=0..80) :
  Y := seq( Re(subs(SOL,h=Pi/40,y(k))),k=0..80) :
> plot( [seq([X[k],Y[k]],k=1..80)], style=point ) ;
```



Solve recurrence using Z-transform

```
> ZREC1 := ztrans(REC1,k,z) ;
  ZREC2 := ztrans(REC2,k,z) ;
```

$$ZREC1 := z\, ztrans(x(k), k, z) - x(0)\, z = ztrans(x(k), k, z) - h\, ztrans(y(k), k, z)$$

$$ZREC2 := z\, ztrans(y(k), k, z) - y(0)\, z = ztrans(y(k), k, z) + h\, ztrans(x(k), k, z) \quad (1.4)$$

Apply initial condizions

```
> ZREC1_WITH_INI := subs(INI,ZREC1) ;
  ZREC2_WITH_INI := subs(INI,ZREC2) ;
```

$$ZREC1\_WITH\_INI := z\, ztrans(x(k), k, z) - z = ztrans(x(k), k, z) - h\, ztrans(y(k), k, z)$$

$$ZREC2\_WITH\_INI := z\, ztrans(y(k), k, z) = ztrans(y(k), k, z) + h\, ztrans(x(k), k, z) \quad (1.5)$$

Solve recurrence in z

```
> SOLZ := solve( {ZREC1_WITH_INI,ZREC2_WITH_INI},{ztrans(x(k), k,
  z),ztrans(y(k), k, z)} ) ;
```

$$SOLZ := \left\{ ztrans(x(k), k, z) = \frac{(z-1)\, z}{h^2 + z^2 - 2\, z + 1}, ztrans(y(k), k, z) = \frac{h\, z}{h^2 + z^2 - 2\, z + 1} \right\} \quad (1.6)$$

Try to match the polinomial z^2-2*z*a*cos(omega)+a^2 with $h^2 + z^2 - 2z + 1$

```
> POLY  := z^2-2*z*a*cos(omega)+a^2 ;
  POLY2 := h^2+z^2-2*z+1 ;
```

$$POLY := z^2 - 2 z a \cos(\omega) + a^2$$

$$POLY2 := h^2 + z^2 - 2 z + 1 \quad (1.7)$$

```
> EQ1 := subs(z=0,POLY-POLY2) ;
  EQ2 := subs(z=0,diff(POLY-POLY2,z)) ;
  EQ3 := subs(z=0,diff(POLY-POLY2,z,z)) ;
```

$$EQ1 := -1 + a^2 - h^2$$

$$EQ2 := -2 a \cos(\omega) + 2$$

$$EQ3 := 0 \quad (1.8)$$

```
> solve( {EQ1,EQ2},{a,omega} ); RES := allvalues(%) ;
```

$$\left\{a = RootOf\left(-1 + \_Z^2 - h^2\right), \omega = \arccos\left(\frac{RootOf\left(-1 + \_Z^2 - h^2\right)}{h^2 + 1}\right)\right\}$$

$$RES := \left\{a = \sqrt{h^2 + 1}, \omega = \arccos\left(\frac{1}{\sqrt{h^2 + 1}}\right)\right\}, \left\{a = -\sqrt{h^2 + 1}, \omega = \pi - \arccos\left(\frac{1}{\sqrt{h^2 + 1}}\right)\right\} \quad (1.9)$$

Get the table of Z transform

```
> T1 := simplify(ztrans( a^k*cos(omega*k), k, z )) ;
  T2 := simplify(ztrans( a^k*sin(omega*k), k, z )) ;
```

$$T1 := \frac{z\left(-z + a \cos(\omega)\right)}{-z^2 + 2 z a \cos(\omega) - a^2}$$

$$T2 := -\frac{z a \sin(\omega)}{-z^2 + 2 z a \cos(\omega) - a^2} \quad (1.10)$$

Simplify and match

```
> simplify(subs( RES[1], T1 )) ;
  simplify(subs( RES[1], T2 )) assuming h>0 ;
```

$$\frac{(z - 1) z}{h^2 + z^2 - 2 z + 1}$$

$$\frac{h z}{h^2 + z^2 - 2 z + 1} \quad (1.11)$$

Thus the solution is

```
> Xk := subs(RES[1],a^k*cos(omega*k)) ;
  Yk := subs(RES[1],a^k*sin(omega*k)) ;
```

$$Xk := \left(\sqrt{h^2 + 1}\right)^k \cos\left(\arccos\left(\frac{1}{\sqrt{h^2 + 1}}\right) k\right)$$

$$Yk := \left(\sqrt{h^2 + 1}\right)^k \sin\left(\arccos\left(\frac{1}{\sqrt{h^2 + 1}}\right) k\right) \quad (1.12)$$

Notice that

```
> taylor( arccos(1/sqrt(h^2+1)), h=0 ) assuming h>0 ;
```

$$h - \frac{1}{3}h^3 + \frac{1}{5}h^5 + O\left(h^7\right) \quad (1.13)$$

So that

```
> Xk_approx := sqrt(h^2+1)^k*cos(h*k) ;
  Yk_approx := sqrt(h^2+1)^k*sin(h*k) ;
```

$$Xk\_approx := \left(\sqrt{h^2+1}\right)^k \cos(h\,k)$$

$$Yk\_approx := \left(\sqrt{h^2+1}\right)^k \sin(h\,k) \quad (1.14)$$

Which is close to exact solution as h goes to 0

Semi implicit scheme

```
> restart:
> with(plots):
```

Approximate x'=-y, y'=x with Semi-Implicit schemer

```
> REC1 := x(k+1)+h*y(k+1)=x(k) ;
  REC2 := y(k+1)=y(k)+h*x(k) ;
  INI  := x(0)=1, y(0)=0 ;
```

$$REC1 := x(k+1) + h\,y(k+1) = x(k)$$

$$REC2 := y(k+1) = y(k) + h\,x(k)$$

$$INI := x(0) = 1, y(0) = 0 \quad (2.1)$$

```
> SOL := simplify(rsolve( {REC1,REC2,INI},{x,y})) assuming h>0;
```

$$SOL := \left\{ x(k) = -\left(2\left(2^k\left(\frac{1}{-h^2+2+h\sqrt{h^2-4}}\right)^k h + 2^k\left(\frac{1}{-h^2+2+h\sqrt{h^2-4}}\right)^k \sqrt{h^2-4} - \left(-\frac{2}{h^2-2+h\sqrt{h^2-4}}\right)^k h + \left(-\frac{2}{h^2-2+h\sqrt{h^2-4}}\right)^k \sqrt{h^2-4}\right)\right) \Big/ \left(\sqrt{h^2-4}\left(-h^2+2+h\sqrt{h^2-4}\right)\left(h^2-2+h\sqrt{h^2-4}\right)\right), y(k) \right. \quad (2.2)$$

$$= \frac{-2^k \left(\frac{1}{-h^2+2+h\sqrt{h^2-4}}\right)^k + \left(-\frac{2}{h^2-2+h\sqrt{h^2-4}}\right)^k}{\sqrt{h^2-4}}$$

Check the discrete "prime integral" x^2+y^2+h*xy=C

```
> SOLINT   := simplify(simplify(subs(SOL,x(k)^2+y(k)^2+h*x(k)*y(k))
  ),size) ;
  #L       := op(1,op(1,SOLINT));
  #R       := op(1,op(2,SOLINT));
  #SOLINT2 := simplify(L*R);
  #SOLINT3 := (1+h^2)^k ;
```

$$SOLINT := \frac{16\, 2^k \left(\frac{1}{-h^2+2+h\sqrt{h^2-4}}\right)^k \left(-\frac{2}{h^2-2+h\sqrt{h^2-4}}\right)^k}{\left(h^2-2+h\sqrt{h^2-4}\right)^2 \left(-h^2+2+h\sqrt{h^2-4}\right)^2} \quad \textbf{(2.3)}$$

```
> expand((h^2-2+h*sqrt(h^2-4))*(-h^2+2+h*sqrt(h^2-4)));
```

$$-4 \quad \textbf{(2.4)}$$

```
> X := seq( evalf(Re(subs(SOL,h=Pi/40,x(k)))),k=0..80) :
  Y := seq( evalf(Re(subs(SOL,h=Pi/40,y(k)))),k=0..80) :
> A := plot( [seq([X[k],Y[k]],k=1..80)],style=point ) ;
  B := plot( [cos(t),sin(t),t=0..2*Pi] ) ;
  display(A,B,scaling=CONSTRAINED) ;
```

$$A := PLOT(\ldots)$$

$$B := PLOT(\ldots)$$



Solve recurrence using Z-transform

```
> ZREC1 := ztrans(REC1,k,z) ;
  ZREC2 := ztrans(REC2,k,z) ;
```

$$ZREC1 := z\, ztrans(x(k), k, z) - x(0)\, z + h\, (z\, ztrans(y(k), k, z) - y(0)\, z) = ztrans(x(k), k, z)$$

$$ZREC2 := z\, ztrans(y(k), k, z) - y(0)\, z = ztrans(y(k), k, z) + h\, ztrans(x(k), k, z) \quad \textbf{(2.5)}$$

Apply initial condizions

```
> ZREC1_WITH_INI := subs(INI,ZREC1) ;
  ZREC2_WITH_INI := subs(INI,ZREC2) ;
```

$$ZREC1\_WITH\_INI := z\, ztrans(x(k), k, z) - z + h\, z\, ztrans(y(k), k, z) = ztrans(x(k), k, z)$$

$$ZREC2\_WITH\_INI := z\, ztrans(y(k), k, z) = ztrans(y(k), k, z) + h\, ztrans(x(k), k, z) \quad \textbf{(2.6)}$$

Solve recurrence in z

```
> SOLZ := solve( {ZREC1_WITH_INI,ZREC2_WITH_INI},{ztrans(x(k), k,
  z),ztrans(y(k), k, z)} ) ;
```

$$SOLZ := \left\{ztrans(x(k), k, z) = \frac{(z-1)\, z}{h^2 z + z^2 - 2 z + 1}, ztrans(y(k), k, z) \right. \quad \textbf{(2.7)}$$

$$= \frac{h\,z}{h^2 z + z^2 - 2 z + 1}\}$$

```
> {ztrans(x(k), k, z) = (z-1)*z/(h^2+z^2-2*z+1), ztrans(y(k), k, z)
  = h*z/(h^2+z^2-2*z+1)};
```

$$\left\{ztrans(x(k), k, z) = \frac{(z-1)\,z}{h^2 + z^2 - 2 z + 1}, ztrans(y(k), k, z) = \frac{h\,z}{h^2 + z^2 - 2 z + 1}\right\} \quad \textbf{(2.8)}$$

Try to match the polinomial z^2-2*z*a*cos(omega)+a^2 with $h^2 + z^2 - 2 z + 1$

```
> POLY  := z^2-2*z*a*cos(omega)+a^2 ;
  POLY2 := h^2*z+z^2-2*z+1 ;
```

$$POLY := z^2 - 2 z a \cos(\omega) + a^2$$

$$POLY2 := h^2 z + z^2 - 2 z + 1 \quad \textbf{(2.9)}$$

```
> EQ1 := subs(z=0,POLY-POLY2) ;
  EQ2 := subs(z=0,diff(POLY-POLY2,z)) ;
  EQ3 := subs(z=0,diff(POLY-POLY2,z,z)) ;
```

$$EQ1 := -1 + a^2$$

$$EQ2 := -2 a \cos(\omega) - h^2 + 2$$

$$EQ3 := 0 \quad \textbf{(2.10)}$$

```
> RES := solve( {EQ1,EQ2},{a,omega} );
```

$$RES := \left\{a = 1, \omega = \pi - \arccos\left(\frac{1}{2} h^2 - 1\right)\right\}, \left\{a = -1, \omega = \arccos\left(\frac{1}{2} h^2 - 1\right)\right\} \quad \textbf{(2.11)}$$

Get the table of Z transform

```
> T1 := simplify(ztrans( cos(omega*k), k, z )) ;
  T2 := simplify(ztrans( sin(omega*k), k, z )) ;
```

$$T1 := \frac{(-z + \cos(\omega))\,z}{-z^2 + 2 z \cos(\omega) - 1}$$

$$T2 := -\frac{z \sin(\omega)}{-z^2 + 2 z \cos(\omega) - 1} \quad \textbf{(2.12)}$$

Simplify and match

```
> TS1 := simplify(subs( RES[1], T1 )) ;
  TS2 := simplify(subs( RES[1], T2 )) assuming h>0 ;
```

$$TS1 := \frac{1}{2} \frac{(2 z + h^2 - 2)\,z}{h^2 z + z^2 - 2 z + 1}$$

$$TS2 := \frac{1}{2} \frac{z\,h \sqrt{-h^2 + 4}}{h^2 z + z^2 - 2 z + 1} \quad \textbf{(2.13)}$$

```
> denom(simplify(a*TS1 + b*TS2));
```

$$2 h^2 z + 2 z^2 - 4 z + 2 \quad \textbf{(2.14)}$$

```
> EQQ := numer(simplify(a*TS1 + b*TS2))/2-(z-1)*z;
```

$$EQQ := \frac{1}{2} z \left(2 a z + a h^2 - 2 a + b h \sqrt{-h^2 + 4}\right) - (z-1)\,z \quad \textbf{(2.15)}$$

```
> EQ1 := subs(z=0,EQQ) ;
```

```
EQ2 := subs(z=0,diff(EQQ,z)) ;
EQ3 := subs(z=0,diff(EQQ,z,z)) ;
```

$$EQ1 := 0$$

$$EQ2 := 1 + \frac{1}{2} a h^2 - a + \frac{1}{2} b h \sqrt{-h^2 + 4}$$

$$EQ3 := 2 a - 2 \tag{2.16}$$

```
> SOL_FOR_X := solve( {EQ2,EQ3}, {a,b} ) ;
```

$$SOL\_FOR\_X := \left\{ a = 1, b = \frac{\sqrt{-h^2 + 4}\; h}{h^2 - 4} \right\} \tag{2.17}$$

```
> XK := subs( SOL_FOR_X, a*cos(omega*k) + b*sin(omega*k)) ;
```

$$XK := \cos(\omega k) + \frac{\sqrt{-h^2 + 4}\; h \sin(\omega k)}{h^2 - 4} \tag{2.18}$$

```
> denom(simplify(a*TS1 + b*TS2));
```

$$2 h^2 z + 2 z^2 - 4 z + 2 \tag{2.19}$$

```
> EQQ := numer(simplify(a*TS1 + b*TS2))/2-h*z;
```

$$EQQ := \frac{1}{2} z \left(2 a z + a h^2 - 2 a + b h \sqrt{-h^2 + 4}\right) - h z \tag{2.20}$$

```
> EQ1 := subs(z=0,EQQ) ;
  EQ2 := subs(z=0,diff(EQQ,z)) ;
  EQ3 := subs(z=0,diff(EQQ,z,z)) ;
```

$$EQ1 := 0$$

$$EQ2 := \frac{1}{2} a h^2 - a + \frac{1}{2} b h \sqrt{-h^2 + 4} - h$$

$$EQ3 := 2 a \tag{2.21}$$

```
> SOL_FOR_Y := solve( {EQ2,EQ3}, {a,b} ) ;
```

$$SOL\_FOR\_Y := \left\{ a = 0, b = -\frac{2\sqrt{-h^2 + 4}}{h^2 - 4} \right\} \tag{2.22}$$

```
> YK := subs( SOL_FOR_Y, a*cos(omega*k) + b*sin(omega*k)) ;
```

$$YK := -\frac{2\sqrt{-h^2 + 4}\; \sin(\omega k)}{h^2 - 4} \tag{2.23}$$

notice that

```
> taylor( XK, h, 3 ) ;
```

$$\cos(\omega k) - \frac{1}{2} \sin(\omega k)\, h + O(h^3) \tag{2.24}$$

```
> taylor( YK, h, 3 ) ;
```

$$\sin(\omega k) + \frac{1}{8} \sin(\omega k)\, h^2 + O(h^4) \tag{2.25}$$

So that Xk and Yk be close to exact solution as h goes to 0